```
Macaulay2, version 1.6
with packages: ConwayPolynomials, Elimination, IntegralClosure, LLLBases,
  PrimaryDecomposition, ReesAlgebra, TangentCone

i1 : S = QQ[a..d]

o1 = S

o1 : PolynomialRing

i2 : -- important example curve #1: twisted cubic
     -- image P1 --> P3, under (s,t) --> (s3, s2t, st2, t3).
     I3 = monomialCurveIdeal(S,{1,2,3})

             2                     2
o2 = ideal (c  - b*d, b*c - a*d, b  - a*c)

o2 : Ideal of S

i3 : C = res (S^1/I3)

      1      3      2
o3 = S  <-- S  <-- S  <-- 0

     0      1      2      3

o3 : ChainComplex

i4 : C.dd

          1                              3
o4 = 0 : S  <------------------------- S  : 1
               | b2-ac bc-ad c2-bd |

          3                  2
     1 : S  <----------------- S  : 2
               {2} | -c d  |
               {2} | b  -c |
               {2} | -a b  |

          2
     2 : S  <----- 0 : 3
               0

o4 : ChainComplexMap

i5 : -- shorthand: res I means the same thing:
     C1 = res I3

      1      3      2
o5 = S  <-- S  <-- S  <-- 0

     0      1      2      3

o5 : ChainComplex

i6 : hf = hilbertSeries I3

           2     3
     1 - 3T  + 2T
```



```
o6 = -------------
             4
        (1 - T)

o6 : Expression of class Divide

i7 : reduceHilbert hf

      1 + 2T
o7 = --------
            2
     (1 - T)

o7 : Expression of class Divide

i8 : dim I3

o8 = 2

i9 : degree I3

o9 = 3

i10 : -- important example curve #2: rational quartic
      -- image P1 --> P3, under (s,t) --> (s4, s3t, st3, t4).
      I4 = monomialCurveIdeal(S,{1,3,4})

                              3      2     2    2    3    2
o10 = ideal (b*c - a*d, c  - b*d , a*c  - b d, b  - a c)

o10 : Ideal of S

i11 : C = res I4

       1      4      4      1
o11 = S  <-- S  <-- S  <-- S  <-- 0

      0      1      2      3      4

o11 : ChainComplex

i12 : C.dd

           1                                    4
o12 = 0 : S  <------------------------------------ S  : 1
                | bc-ad b3-a2c ac2-b2d c3-bd2 |

           4                              4
      1 : S  <---------------------------- S  : 2
                {2} | -b2 -ac -bd -c2 |
                {3} | c   d   0   0   |
                {3} | a   b   -c  -d  |
                {3} | 0   0   a   b   |

           4                  1
      2 : S  <-------------- S  : 3
                {4} | d  |
                {4} | -c |
                {4} | -b |
                {4} | a  |
```

```
            1
      3 : S  <----- 0 : 4
                 0

o12 : ChainComplexMap

i13 : hf = hilbertSeries I4

           2     3     4    5
      1 - T  - 3T  + 4T  - T
o13 = -----------------------
                      4
              (1 - T)

o13 : Expression of class Divide

i14 : reduceHilbert hf

                    2    3
      1 + 2T + 2T  - T
o14 = -----------------
                   2
           (1 - T)

o14 : Expression of class Divide

i15 : dim I4

o15 = 2

i16 : degree I4

o16 = 4

i17 : -- example: dual of an ideal
      R = S/(a*d-b*c)

o17 = R

o17 : QuotientRing

i18 : I = ideal(a,b)

o18 = ideal (a, b)

o18 : Ideal of R

i19 : C = res I

       1      2      2      2      2      2
o19 = R  <-- R  <-- R  <-- R  <-- R  <-- R

      0      1      2      3      4      5

o19 : ChainComplex

i20 : C.dd

           1               2
```



```
o20 = 0 : R  <----------- R  : 1
                | a b |

            2                     2
      1 : R  <----------------- R  : 2
                {1} | -d -b |
                {1} | c  a  |

            2                     2
      2 : R  <----------------- R  : 3
                {2} | b  -a |
                {2} | -d c  |

            2                   2
      3 : R  <--------------- R  : 4
                {3} | c a |
                {3} | d b |

            2                     2
      4 : R  <----------------- R  : 5
                {4} | b  -a |
                {4} | -d c  |

o20 : ChainComplexMap

i21 : phi = (syz gens I)

o21 = {1} | -d -b |
      {1} | c  a  |

                2       2
o21 : Matrix R  <--- R

i22 : phi' = transpose phi

o22 = {-2} | -d c |
      {-2} | -b a |

                2       2
o22 : Matrix R  <--- R

i23 : ker phi'

o23 = image {-1} | c a |
            {-1} | d b |

                                 2
o23 : R-module, submodule of R

i24 : I' = Hom(I,R) -- the module has 2 generators

o24 = image {-1} | c a |
            {-1} | d b |

                                 2
o24 : R-module, submodule of R

i25 : f = I'_{0}

o25 = | 1 |
```

```
      | 0 |

o25 : Matrix

i26 : super f

o26 = {-1} | c |
      {-1} | d |

              2       1
o26 : Matrix R  <--- R

i27 : super I'_{1} -- identity map

o27 = {-1} | a |
      {-1} | b |

              2       1
o27 : Matrix R  <--- R

i28 : prune I'

o28 = cokernel | b  a  |
               | -d -c |

                           2
o28 : R-module, quotient of R

i29 : -- normal bundle for an elliptic curve in P^3
      use S

o29 = S

o29 : PolynomialRing

i30 : Ie = ideal random(S^1, S^{-2,-2});

o30 : Ideal of S

i31 : netList Ie_*

      +----------------------------------------------------------------+
      |5 2          1 2   7      3      1 2                 4      3 2|
o31 = |-a  + 4a*b + -b  + -a*c + -b*c + -c  + 3a*d + 2b*d + -c*d + -d |
      |2            2     4      5      4                   3      2  |
      +----------------------------------------------------------------+
      |2 2   7       2   4      4       2                     4 2  |
      |-a  + -a*b + b  + -a*c + -b*c + c  + 2a*d + 9b*d + 4c*d + -d   |
      |9     2           3      5                             7    |
      +----------------------------------------------------------------+

i32 : codim Ie

o32 = 2

i33 : degree Ie

o33 = 4

i34 : genus Ie
```



```
o34 = 1

i35 : N = prune Hom(Ie, S^1/Ie);

i36 : R = S/Ie

o36 = R

o36 : QuotientRing

i37 : N = N ** R

       2
o37 = R

o37 : R-module, free, degrees {2:-2}

i38 : X = variety Ie

o38 = X

o38 : ProjectiveVariety

i39 : sheaf N -- So N = OO_X(2) ++ OO_X(2)

         2
o39 = OO  (2)
        X

o39 : coherent sheaf on X, free
```